// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/rewards-booster/interfaces/IOracle.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

/**
 * @title Asymetrix Protocol V2 IOracle
 * @author Asymetrix Protocol Inc Team
 * @notice An interface that all custom oracles should implement.
 */
interface IOracle {
    /**
     * @notice Returns the latest answer.
     * @return _answer The latest answer.
     */
    function latestAnswer() external view returns (int256 _answer);

    /**
     * @notice Returns the data from the latest round.
     * @return _roundId The round ID.
     * @return _answer The answer from the latest round.
     * @return _startedAt Timestamp of when the round started.
     * @return _updatedAt Timestamp of when the round was updated.
     * @return _answeredInRound Deprecated. Previously used when answers could take multiple rounds to be computed.
     */
    function latestRoundData()
        external
        view
        returns (uint80 _roundId, int256 _answer, uint256 _startedAt, uint256 _updatedAt, uint80 _answeredInRound);

    /**
     * @notice Returns the number of decimals in the answer.
     * @return The number of decimals in the answer.
     */
    function decimals() external pure returns (uint8);
}

// File: contracts/rewards-booster/libraries/RewardsBoosterErrors.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

/**
 * @title Asymetrix Protocol V2 RewardsBoosterErrors library
 * @author Asymetrix Protocol Inc Team
 * @notice A library with errors for rewards booster contracts.
 */
library RewardsBoosterErrors {
    error WrongLockDurtionSettingsNumber();
    error WrongBalancerPoolTokensNumber();
    error WrongLockDurtionSettings();
    error TooSmallBoostThreshold();
    error NoEmptySlotsInThisPool();
    error InvalidStakeArguments();
    error WrongValidityDuration();
    error TooBigBoostThreshold();
    error TooMuchLocksCreated();
    error StakeTokenWithdraw();
    error LockIsNotFinished();
    error WrongLockDurtion();
    error NotExistingPool();
    error NotExistingLock();
    error LengthsMismatch();
    error NotContract();
    error ZeroAddress();
    error ZeroAmount();
    error StalePrice();
    error WrongTick();
    error WrongTokensRatio();
    error WrongMaxTokenDominance();
    error StubMethod();
}

// File: contracts/rewards-booster/oracles/OracleBalancerWeighted.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

import { RewardsBoosterErrors } from "../libraries/RewardsBoosterErrors.sol";
import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import { IVault } from "../../vendor/balancer-v2/interfaces/IVault.sol";
import { IPool } from "../../vendor/balancer-v2/interfaces/IPool.sol";
import { Ownable } from "@openzeppelin/contracts/access/Ownable.sol";
import { Address } from "@openzeppelin/contracts/utils/Address.sol";
import { IOracle } from "../interfaces/IOracle.sol";

/**
 * @title Asymetrix Protocol V2 OracleBalancerWeighted
 * @author Asymetrix Protocol Inc Team
 * @notice An oracle for a token that uses Balancer weighted pool to price it. Only pools with 2 tokens are supported by
 *         this oracle. One of them must be WETH.
 */
contract OracleBalancerWeighted is Ownable, IOracle {
    using Address for address;

    IVault private vault;
    IPool private pool;
    IOracle private ethOracle;
    uint256 private tokenIndex;
    uint256 private wethIndex;
    uint32 private validityDuration;

    /**
     * @notice Deploy the OracleBalancerWeighted contract.
     * @param _vault Balancer Vault contract address.
     * @param _pool Balancer Pool contract address.
     * @param _ethOracle ETH Oracle contract address.
     * @param _validityDuration A duration (in seconds) during which the latest answer from the Chainlink oracle is
     *                          valid.
     */
    constructor(address _vault, address _pool, address _ethOracle, address _weth, uint32 _validityDuration) {
        onlyContract(_vault);
        onlyContract(_pool);
        onlyContract(_ethOracle);
        onlyContract(_weth);

        vault = IVault(_vault);
        pool = IPool(_pool);
        ethOracle = IOracle(_ethOracle);

        setNewValidityDuration(_validityDuration);

        (IERC20[] memory _tokens, , ) = vault.getPoolTokens(pool.getPoolId());

        if (_tokens.length > 2) revert RewardsBoosterErrors.WrongBalancerPoolTokensNumber();

        address(_tokens[0]) == _weth ? tokenIndex = 1 : wethIndex = 1;
    }

    /**
     * @notice Sets a new validity duration (in seconds) by an owner.
     * @param _newValidityDuration A new duration (in seconds) during which the latest answer from the Chainlink oracle
     *                             is valid.
     */
    function setValidityDuration(uint32 _newValidityDuration) external onlyOwner {
        setNewValidityDuration(_newValidityDuration);
    }

    /// @inheritdoc IOracle
    function latestAnswer() external view returns (int256) {
        return getLatestAnswer();
    }

    /// @inheritdoc IOracle
    function latestRoundData()
        external
        view
        returns (uint80 _roundId, int256 _answer, uint256 _startedAt, uint256 _updatedAt, uint80 _answeredInRound)
    {
        return (0, getLatestAnswer(), block.timestamp, block.timestamp, 0);
    }

    /**
     * @notice Returns Balancer Vault address.
     * @return Balancer Vault address.
     */
    function getVault() external view returns (IVault) {
        return vault;
    }

    /**
     * @notice Returns Balancer Pool address.
     * @return Balancer Pool address.
     */
    function getPool() external view returns (IPool) {
        return pool;
    }

    /**
     * @notice Returns ETH Oracle address.
     * @return ETH Oracle address.
     */
    function getEthOracle() external view returns (IOracle) {
        return ethOracle;
    }

    /**
     * @notice Returns validity duration in seconds.
     * @return Validity duration in seconds.
     */
    function getValidityDuration() external view returns (uint32) {
        return validityDuration;
    }

    /// @inheritdoc IOracle
    function decimals() external pure returns (uint8) {
        return 18;
    }

    /**
     * @notice Sets a new validity duration (in seconds).
     * @param _newValidityDuration A new duration (in seconds) during which the latest answer from the Chainlink oracle
     *                             is valid.
     */
    function setNewValidityDuration(uint32 _newValidityDuration) private {
        if (_newValidityDuration == 0) revert RewardsBoosterErrors.WrongValidityDuration();

        validityDuration = _newValidityDuration;
    }

    /**
     * @notice Checks if an address is a contract.
     * @param _contract An address to check.
     */
    function onlyContract(address _contract) private view {
        if (!_contract.isContract()) revert RewardsBoosterErrors.NotContract();
    }

    /**
     * @notice Returns the latest answer.
     * @return The latest answer.
     */
    function getLatestAnswer() private view returns (int256) {
        IOracle _ethOracle = ethOracle;
        (, int256 _answer, , uint256 _updatedAt, ) = _ethOracle.latestRoundData();

        if (block.timestamp - _updatedAt > validityDuration) revert RewardsBoosterErrors.StalePrice();

        uint256 _ethPrice = (uint256(_answer) * 1e18) / 10 ** _ethOracle.decimals();
        IPool _pool = pool;
        (, uint256[] memory _balances, ) = vault.getPoolTokens(_pool.getPoolId());
        uint256[] memory _weights = _pool.getNormalizedWeights();
        uint256 _tokenIndex = tokenIndex;
        uint256 _wethIndex = wethIndex;
        uint256 _tokenPrice = (((_balances[_tokenIndex] * 1e18) / _weights[_tokenIndex]) * 1e18) /
            ((_balances[_wethIndex] * 1e18) / _weights[_wethIndex]);

        return int256((_ethPrice * 1e18) / _tokenPrice);
    }
}

// File: contracts/vendor/balancer-v2/interfaces/IPool.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

interface IPool {
    function getPoolId() external view returns (bytes32);

    function getNormalizedWeights() external view returns (uint256[] memory);

    function totalSupply() external view returns (uint256);

    function balanceOf(address _user) external view returns (uint256);
}

// File: contracts/vendor/balancer-v2/interfaces/IVault.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

import { IERC20 } from "@openzeppelin/contracts/token/ERC20/IERC20.sol";

interface IVault {
    function getPoolTokens(
        bytes32 _poolId
    ) external view returns (IERC20[] memory _tokens, uint256[] memory _balances, uint256 _lastChangeBlock);
}